10 Difference Lists

A prolog list is accessed through its head and its tail. The setback of this way of viewing the list is that when we have to access the nth element, we must access all the elements before it first. If, for example, we need to add an element at the end of the list, we must go through all the elements in the list to reach that element.

```
add(X, [H | T], [H | R]) :- add(X, T, R).
add(X, [], [X]).
```

There is an alternative technique of representing lists in prolog that lets us access the end of a list easier. A difference list is represented by two parts, the start of the list S and the end of the list E:

```
S: [1,2,3,4]E: [3,4]S-E: [1,2]S-E(the difference list) represents the list obtained by removing part E from part S:
```

There are no advantages when using difference lists like in the previous example, but when combined with the concepts of free variables and unification, difference lists become a powerful tool. For example, list [1,2] can be represented by the difference list [1,2|X]-X, where X is a free variable. We can write the add predicate with difference lists in the following way:

```
add(X,LS,LE,RS,RE):-RS=LS,LE=[X | RE].
We test it in Prolog by asking the following query:
?- LS=[1,2,3,4 | LE],add(5,LS,LE,RS,RE).
LE = [5 | RE],
LS = [1,2,3,4,5 | RE],
RS = [1,2,3,4,5 | RE]?
yes
```

To better understand the way the add predicate works, we can imagine the list is represented by two "pointers", one pointing to the start of the list (LS) and the second one to the end of the list (LE), a variable without an assigned value.



Figure 10.1 – Adding an element at the end of a difference list

The result is also represented by two "pointers". The resulting list will be the input list with the new element inserted at the end. The beginning of the input and result lists is the same so we can unify the result list start variable with the input list start variable (RS=LS).

The result list must end, just like the input list, in a variable (RE), but we must, somehow, modify the input list to add the new element at the end. Because the end of the input list is a free variable, we can unify it with the list beginning with the new element followed by a new variable, the new end of the list (LE=[X | RE]). After the predicate finished execution we can see that the input list LS and result list RS have the same values, but the end of the input list is no longer a variable (LE=[5 | RE]).

10.1 Tree traversal

The tree traversal predicates are used to extract the elements in a tree to a list in a specific order. The computation intensive part of these predicates is not the traversing itself but the combination of the result lists to obtain the final result. Although hidden from us, prolog will go through the same elements of a list many times to form the result list. We can save a lot of work by changing regular lists to difference lists.

The inorder predicate using a regular list to store the result:

inorder(t(K,L,R),List):- inorder(L,ListL), inorder(R,ListR), append1(ListL,[K | ListR],List). inorder (nil,[]).

By executing a query trace on the inorder predicate we can easily observe the amount of work performed by the append predicate. It is also visible that the append predicate will access the same elements in the result list more than once as the intermediary results are appended to obtain the final result:

[...]

- 8 3 Exit: inorder(t(5,nil,nil),[5])?
- 12 3 Call: append1([2],[4,5],_1594)?
- 13 4 Call: append1([],[4,5],_10465) ?
- 13 4 Exit: append1([],[4,5],[4,5])?
- 12 3 Exit: append1([2],[4,5],[2,4,5]) ?

[...]

- 22 2 Call: append1([2,4,5],[6,7,9],_440)?
- 23 3 Call: append1([4,5],[6,7,9],_20633) ?
- 24 4 Call: append1([5],[6,7,9],_21109) ?
- 25 5 Call: append1([],[6,7,9],_21585) ?
- 25 5 Exit: append1([],[6,7,9],[6,7,9])?
- 24 4 Exit: append1([5],[6,7,9],[5,6,7,9]) ?
- 23 3 Exit: append1([4,5],[6,7,9],[4,5,6,7,9])?

22 2 Exit: append1([2,4,5],[6,7,9],[2,4,5,6,7,9]) ? [...]

We can improve the efficiency of the inorder predicate by rewriting it using difference lists. The inorder_dl predicate has 3 parameters: the tree node it is currently processing, the start of the result list and the end of the result list:

/* when we reached the end of the tree we unify the beggining and end of the partial
result list - representing an empty list as a difference list */
inorder_dl(nil,L,L).
inorder_dl(t(K,L,R),LS,LE):/* obtain the start and end of the lists for the left and right subtrees */
inorder_dl(L,LSL,LEL),
inorder_dl(R,LSR,LER),
/* the start of the result list is the start of the left subtree list */
LS=LSL,
/* insert the key between the end of the left subtree list and start of the right subtree list */
LEL=[K | LSR],
/* the end of the result list is the end of the right subtree list */
LE=LER.



Figure 10.2 – Appending two difference lists

The predicate can be simplified by replacing the explicit unifications with implicit unifications:

inorder_dl(nil,L,L). inorder_dl(t(K,L,R),LS,LE):-inorder_dl(L,LS,[K | LT]), inorder_dl(R,LT,LE).

Exercise 10.1: Study the execution of the following queries:

- ?- tree1(T), inorder_dl(T,L,[]).
- ?- tree1(T), inorder_dl(T,L,_).

Exercise 10.2: *Implement the* preorder_dl *tree traversal predicate using difference lists.*

Exercise 10.3: *Implement the* postorder_dl *tree traversal predicate using difference lists.*

10.2 Sorting – quicksort

Remember the *quicksort* algorithm (and predicate): the input sequence is divided in two parts – the sequence of elements smaller or equal to the pivot and the sequence of elements larger than the pivot; the procedure is called recursively on each partition, and the resulting sorted sequences are appended together with the pivot to generate the sorted sequence:

```
quicksort([H | T], R):-
    partition(H, T, Sm, Lg),
    quicksort(Sm, SmS),
    quicksort(Lg, LgS),
    append(SmS, [H | LgS], R).
quicksort([], []).
```

Just as for the inorder predicate, the quicksort predicate will waste a lot of execution time to append the results of the recursive calls. To avoid this we can apply difference lists again:

```
quicksort_dl([H | T],S,E):-
    partition(H,T,Sm,Lg),
    quicksort_dl(Sm,S,[H | L]),
    quicksort_dl(Lg,L,E).
quicksort_dl([],L,L).
```

The partition predicate is the same as the one for the old quicksort predicate, its purpose being to divide the list in two by comparing each element with the pivot. All elements in the list must be accessed for this operation so we cannot improve the performance for the partition predicate.

The quicksort predicate works in the same way as before: divides the list in elements larger and smaller than the pivot element and applies quicksort recursively on the each partition. The difference from the original version is the result list, represented by two elements, the start and the end of the list, and, consequently, the way in which the results of the two recursive calls are put together with the pivot (figure 10.3 below).

Exercise 10.4: Study the execution of the following queries:

- ?- quicksort_dl([4,2,5,1,3],L,[]).
- ?- quicksort_dl([4,2,5,1,3],L,_).



Figure 10.3 – Quicksort with difference lists

10.3 Quiz exercises

10.3.1 Write a predicate which transforms an incomplete list intro a difference list (and one which makes the opposite transformation).

10.3.2. Write a predicate which transforms a complete list into a difference list (and one which makes the opposite transformation).

10.4 Problems

10.4.1 Write a predicate which flattens a deep list using difference lists instead of append.

10.4.2 Write a predicate which collects all even keys in a binary tree, using difference lists.

10.4.3. Write a predicate which collects, from a binary incomplete search tree, all keys between K1 and K2, using difference lists.